

Информатика

Ткаченко Кирилл Станиславович

*Инженер 1-й категории,
ФГАОУ ВО «Севастопольский
государственный университет»*



Программная реализация демонстрации «египетского» деления на школьном алгоритмическом языке

В настоящей статье рассматривается программная реализация демонстрации «египетского» деления на школьном алгоритмическом языке. Приводится полный исходный текст и результаты работы программы. Разработка будет полезна школьникам, увлекающимся математикой, изучающим программирование и школьный алгоритмический язык.

Изучение информатики, программирования должно быть интересным. Некоторый интерес представляет программирование занимательных и исторических математических задач. К таким задачам можно отнести и исторические методы умножения и деления. На известном хрестоматийном сайте по программированию рассматривается демонстрация так называемого «египетского» деления на различных языках программирования [1].

Целью настоящей статьи является рассмотрение программной реализации демонстрации «египетского» деления на школьном алгоритмическом языке.

«Египетское» деление удобно объяснить на примере нахождения частного и остатка от целочисленного деления. Пусть необходимо разделить 580 на 34.

Вначале строится таблица, в которой записываются степени двойки и удвоения делителя, пока удвоения делителя меньше делимого:

Степень	Двойки	Удвоение
	1	34
	2	68
	4	136
	8	272
	16	544

Затем формируется таблица, в которой, в обратном порядке, записываются эти степени двойки, удвоения, ответ, равный сумме «помещающихся» степеней двойки, и накопленное значение (накопитель), равное сумме соответствующих им удвоений:

СтепеньДвойки	Удвоение	Ответ	Накопитель
16	544	16	544
8	272	16	544
4	136	16	544
2	68	16	544
1	34	17	578

Получаются частное (равное ответу) и остаток (равный разности между делимым и накопителем):

Частное	Остаток
17	2

Этот же пример ниже будет демонстрировать программа на школьном алгоритмическом языке.

Главной алгоритм Потенциал вызывает алгоритм для демонстрации «египетского» деления:

```
алг Потенциал
нач
```

```
    ЕгипетскоеДеление(580, 34);
```

```
кон
```

Алгоритм ЕгипетскоеДеление(цел Делимое, цел Делитель) выполняет непосредственную демонстрацию процесса. Целочисленные его аргументы Делимое и Делитель содержат нужные для деления делимое и делитель. В целочисленных массивах СтепениДвойки и Удвоения находятся требуемые для процесса, рассчитываемые степени двойки и удвоения делителя, в целочисленных переменных Степень, СтепеньДвойки, Удвоение находятся текущие значения показателя степени, значения степени двойки и удвоения делителя, Частное и Остаток — результирующие частное и остаток, Ответ и Накопитель — их промежуточные значения:

```
алг ЕгипетскоеДеление(цел Делимое, цел Делитель)
```

```
нач
```

```
    цел таб СтепениДвойки[0 : 15], Удвоения[0 : 15];
```

```
    цел Степень, СтепеньДвойки, Удвоение;
```

```
    цел Частное, Остаток;
```

```
    цел Ответ, Накопитель;
```

Вначале степень двойки равна единице (20), текущее значение удвоения равно делителю, показатель степени нулевой, также выводится заголовок таблицы:

```
    СтепеньДвойки := 1;
```

```
    Удвоение := Делитель;
```

```
    Степень := 0;
```

```
    вывод "СтепеньДвойки Удвоение", нс;
```

Пока текущее значение удвоения меньше делимого:

```
нц пока Удвоение <= Делимое
```

Заполняются таблицы, хранящие степень двойки и удвоения:

```
    СтепениДвойки[Степень] := СтепеньДвойки;
```

```
    Удвоения[Степень] := Удвоение;
```

Эти значения выводятся на стандартное устройство вывода:

```
    вывод СтепеньДвойки : 13, Удвоение : 9, нс;
```

Формируются новые значения степени двойки, удвоения и показателя степени:

```
СтепеньДвойки := 2 * СтепеньДвойки;
```

```
Удвоение := 2 * Удвоение;
```

```
Степень := Степень + 1;
```

кц

Отображается заголовок таблицы, счет производится с предпоследнего достигнутого значения показателя степени, ответ и накопитель нулевые:

```
вывод "СтепеньДвойки Удвоение Ответ Накопитель", нс;
```

```
Степень := Степень - 1;
```

```
Ответ := 0;
```

```
Накопитель := 0;
```

Пока показатель степени не меньше нуля и накопленное значение меньше делимого:

```
нц пока Степень >= 0 и Накопитель < Делимое
```

Если сумма на накопителе и результат удвоения не превышают делимое, то накопитель увеличивается на результат удвоения, ответ увеличивается на степень двойки:

```
если Накопитель + Удвоения[Степень] <= Делимое
```

```
то
```

```
    Накопитель := Накопитель + Удвоения[Степень];
```

```
    Ответ := Ответ + СтепениДвойки[Степень];
```

```
все
```

Отображается текущее табличное значение степени двойки, удвоения, ответа и накопленное значение, уменьшается значение показателя степени:

```
вывод СтепениДвойки[Степень] : 13, Удвоения[Степень] : 9, Ответ : 6, Накопитель : 11, нс;
```

```
Степень := Степень - 1;
```

кц

Частное равно рассчитанному ответу, если накопленная величина меньше делимого, то остаток равен разности делимого и накопленной величины, иначе он равен нулю:

```
Частное := Ответ;
```

```
если Накопитель < Делимое
```

```
то
```

```
    Остаток := Делимое - Накопитель;
```

```
иначе
```

```
    Остаток := 0;
```


```
все
```

Отображаются в табличное форме частное и остаток от целочисленного деления:

```
вывод "Частное Остаток", нс;
```

```
вывод Частное : 7, Остаток : 8, нс;
```

кон



Полный исходный текст программы находится в приложении А, результаты работы программы — Б и на снимке экрана.

Полученная программа может быть полезна школьникам, и всем, увлекающимся азами программирования, занимательными математическими задачами, изучающими школьный алгоритмический язык и основные конструкции других языков программирования.

Список литературы

1. Egyptian division // Rosetta Code URL: http://rosettacode.org/wiki/Egyptian_division (дата обращения: 19.10.2020).

Приложение А

Полный исходный текст программы на школьном алгоритмическом языке

```
алг Потенциал
нач
    ЕгипетскоеДеление(580, 34);
кон

алг ЕгипетскоеДеление(цел Делимое, цел Делитель)
нач
    цел таб СтепениДвойки[0 : 15], Удвоения[0 : 15];
    цел Степень, СтепеньДвойки, Удвоение;
    цел Частное, Остаток;
    цел Ответ, Накопитель;

    СтепеньДвойки := 1;
    Удвоение := Делитель;
    Степень := 0;
    вывод "СтепеньДвойки Удвоение", нс;

    нц пока Удвоение <= Делимое

        СтепениДвойки[Степень] := СтепеньДвойки;
        Удвоения[Степень] := Удвоение;

        вывод СтепеньДвойки : 13, Удвоение : 9, нс;

        СтепеньДвойки := 2 * СтепеньДвойки;
        Удвоение := 2 * Удвоение;
        Степень := Степень + 1;
    кц

    вывод "СтепеньДвойки Удвоение Ответ Накопитель", нс;
    Степень := Степень - 1;
    Ответ := 0;
    Накопитель := 0;
```

```
нц пока Степень >= 0 и Накопитель < Делимое
  если Накопитель + Удвоения[Степень] <= Делимое
    то
      Накопитель := Накопитель + Удвоения[Степень];
      Ответ := Ответ + СтепениДвойки[Степень];
  все
```

```
  вывод СтепениДвойки[Степень] : 13, Удвоения[Степень] : 9,
  Ответ : 6, Накопитель : 11, нс;
  Степень := Степень - 1;
кц
```

```
Частное := Ответ;
если Накопитель < Делимое
  то
    Остаток := Делимое - Накопитель;
  иначе
    Остаток := 0;
все
```

```
вывод "Частное Остаток", нс;
вывод Частное : 7, Остаток : 8, нс;
кон
```

Приложение Б

Результаты работы программы

СтепеньДвойки	Удвоение
---------------	----------

1	34
2	68
4	136
8	272
16	544

СтепеньДвойки	Удвоение	Ответ	Накопитель
---------------	----------	-------	------------

16	544	16	544
8	272	16	544
4	136	16	544
2	68	16	544
1	34	17	578

Частное	Остаток
---------	---------

17	2
----	---